
	Reason for change:
(

	1)The Periodic RA Update Timer is started after READY timer is stopped or expired. As the Network does not include any READY Timer Value in its answer in RAU_ACC msg, then the value proposed by the MS or default value of Ready Timer shall be applied as per 24.008 section 4.7.2.1.1.
2)  24.008  9.4.14.5:  P-TMSI IE of RAU REQ should only be included in Iu mode only.

A prose CR will be raised at next RAN5#57 meeting.
3)  Initialisation of the PLMN, RAC, LAC is done for GERAN cell Cell26 which is not used in the testcase. It should be done with used geran Cell24.

4)  Incorrect condition for checking ISR support at step 21.  

5)  The ciphering of RAU_ACCEPT msg at step 1 is not correct as the security procedure needs to be performed.

A prose CR will be raised at next RAN5#57 meeting.
6)  The UE sends RAU REQ with PDP context status. As per 44.065 section 6.8   XID procedure is needed.

7)  There is wrong Update_result in RAU_ACC in case PS_only attached MS.

     v_UpdateType := f_UG_RauReq_GetUpdateValueForRauAcc  

     causes to mix UpdateType and Update_result what is wrong.

8)  36.523-1 Table 9.2.3.3.5.3.3-8 not correct implemented for PS_only attached MS in the Step 9b3.
9)  All steps names in comments are not according Table 9.2.3.3.5.3.2-1: eg. “1a2” should  be “1b2”.

	
	

	Summary of change:
(

	1)  Define default value of READY timer = 44s as per 24.008 section 11.2.2.
   - Read the READY timer value from RAU_request.

    - Introduce new function f_Get_GPRS_Timer to calculate delay for Ready timer requested by MS. 

    - Apply the delay of MS requested READY timer after RAU_ACCEPT msg.

2.) Removed the checking of P-TMSI in RAU_REQ msg. 

3) Change the GERAN cell from geran_Cell26 to geran_Cell24 as shown below.
4) Corrected the checking condition for ISR as shown below.
5) Added Authentication Procedure in RAU procedure and 

    initialized all security information.   v_SecurityInfo := f_GERAN_SecurityInfo_Get(); 
6) Added XID procedure after first RAU while PDP context status is not deactivated.  

7) Function f_UG_RauReq_GetUpdateValueForRauAcc returns Update result so 

    Do not use it for Update type. Update type taken from receive msg RAU_REQ.
8) Change the condition for setting Update_result at step 9b3
9) Correct steps names in comments.

	Other comments:

Change 1

	Testcase name
	9.2.3.3.5

	Reason for change
	 1)The Periodic RA Update Timer is started after READY timer is stopped or expired. As the Network does not include any READY Timer Value in its answer in RAU_ACC msg, then the value proposed by the MS or default value of Ready Timer shall be applied as per 24.008 section 4.7.2.1.1.
2)  24.008  9.4.14.5:  P-TMSI IE of RAU REQ should only be included in Iu mode only.

A prose CR will be raised at next RAN5#57 meeting.
3)  Initialisation of the PLMN, RAC, LAC is done for GERAN cell Cell26 which is not used in the testcase. It should be done with used geran Cell24.

4)  Incorrect condition for checking ISR support at step 21.  

5)  The ciphering of RAU_ACCEPT msg at step 1 is not correct as the security procedure needs to be performed.

A prose CR will be raised at next RAN5#57 meeting.
6)  The UE sends RAU REQ with PDP context status. As per 44.065 section 6.8   XID procedure is needed.

7)  There is wrong Update_result in RAU_ACC in case PS_only attached MS.

     v_UpdateType := f_UG_RauReq_GetUpdateValueForRauAcc  

     causes to mix UpdateType and Update_result what is wrong.

8)  36.523-1 Table 9.2.3.3.5.3.3-8 not correct implemented for PS_only attached MS in the Step 9b3.
9)  All steps names in comments are not according Table 9.2.3.3.5.3.2-1: eg. “1a2” should  be “1b2”



	Summary of change
	1)  Define default value of READY timer = 44s as per 24.008 section 11.2.2.
   - Read the READY timer value from RAU_request.

    - Introduce new function f_Get_GPRS_Timer to calculate delay for Ready timer requested by MS. 

    - Apply the delay of MS requested READY timer after RAU_ACCEPT msg.

2.) Removed the checking of P-TMSI in RAU_REQ msg. 

3) Change the GERAN cell from geran_Cell26 to geran_Cell24 as shown below.
4) Corrected the checking condition for ISR as shown below.
5) Added Authentication Procedure in RAU procedure and 

    initialized all security information.   v_SecurityInfo := f_GERAN_SecurityInfo_Get(); 
6) Added XID procedure after first RAU while PDP context status is not deactivated.  

7) Function f_UG_RauReq_GetUpdateValueForRauAcc returns Update result so 

    Do not use it for Update type. Update type taken from receive msg RAU_REQ.
8) Change the condition for setting Update_result at step 9b3
9) Correct steps names in comments.

	Source of change
	9_2\NAS_TrackingArea_UG_GERAN.ttcn

	MCC160 Comment
	1) Whatever change is made will need to be documented in the prose.  Therefore this will be implemented according to the RAN5 decision.  Preferred options are either: 
a) Specify value of 0 for READY timer in Accept message. 

b) Don’t include READY timer value in Accept message but update prose to take default value (44s) into account

New GPRS Timer function not needed
2 – 9) Accepted


Before change:

	  function f_TC_9_2_3_3_5_GERAN () runs on GERAN_PTC

  { /* Periodic Routing Area Update (multi RAT test) */

    /* Test Purpose (see f_TC_9_2_3_3_5_EUTRA) */

    /* @sic R5-113805: pc_GERAN and px_RATComb_Tested are checked at top-level sic@ */

    var octetstring v_PLMN;

    var O1_Type v_RAC;

    var O2_Type v_LAC;

    var G_LLC_DATAMESSAGE_IND v_LLCDataInd;

    var B3_Type v_UpdateType, v_UpdateResult;

    var template (value) LLCUnitDataReqType v_RAUAccept;

    var template UE_NetworkCap v_UE_NetworkCap := cr_UENetworkCap_Any ('58'O) ifpresent; // @sic R5s120210 Subject to prose CR approval sic@

    var IRAT_Coordination_MSG v_RxdMsg;

    var GERAN_SecurityInfo_Type v_SecurityInfo;

    var CoOrd_GUTIParam_Type v_GUTI;

    var bitstring v_PTMSI;

    timer t_T3323 := f_UG_SetTimerToleranceMax (7.0 * 60.0); // @sic R5-120768, R5s120210 sic@

    timer t_T3312 := f_UG_SetTimerToleranceMin (7.0 * 60.0);  // Periodic RA Update timer - 10%

    f_GERAN_Init(EUTRA_GERAN); // This is still only a 2 PTC test

    v_PLMN := f_GERAN_PLMN_Get(geran_Cell26);

    v_RAC := f_GERAN_RAC_Get(geran_Cell26);

    v_LAC := f_GERAN_LAC_Get(geran_Cell26);

    f_GERAN_CreateCell (geran_Cell24);

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);  // @sic R5s110777 sic@

    f_GERAN_Pre_Preamble (geran_Cell24, NotSpecified); // @sic R5s110176 sic@

    EUTRA.receive (cmr_IRAT_AuthInfo) -> value v_RxdMsg;

    v_SecurityInfo.authPS := v_RxdMsg.Authentication.AuthPS;

    v_GUTI := v_RxdMsg.Authentication.GutiParams;

    v_PTMSI := '11'B & substr (v_GUTI.M_TMSI, 2, 6) & v_GUTI.MME_Code &  substr (v_GUTI.M_TMSI, 16, 16);

    f_GERAN_SecurityInfo_Set (v_SecurityInfo);

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_NonSuitableCell);

    f_GERAN_TestBody_Set( true );

    //@siclog "Step 1a1" siclog@

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_ServingCell);

    // Routing Area Update on GERAN side

    //@siclog "Step 1a2" siclog@

    v_LLCDataInd := f_GPRS_ReceiveUplinkMsg (geran_Cell24, car_G_LLC_UnitData_IND ( cr_G_RAUReq_Ids ( cr_GMM_UpdateType,  // @sic R5-113809 sic@

                                                                                                      cr_RAI (omit),

                                                                                                      cr_PTMSI_SignatureAny ifpresent,

                                                                                                      ?,

                                                                                                      cdr_MobileIdTMSI_tlv(bit2oct(v_PTMSI)), // @sic R5-121764 sic@

                                                                                                      cr_MobileIdAny('1A'O) ifpresent,

                                                                                                      cr_RoutingAreaId2 ifpresent)));

    // Check ISR support

    if (v_LLCDataInd.unit.msg.rauReq.msNetworkCap.isrSupport != '1'B) {

      f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 1a2");

    }

    v_UpdateType := f_UG_RauReq_GetUpdateValueForRauAcc(v_LLCDataInd.unit.msg.rauReq, '100'B);

    if (v_UpdateType == '000'B) { // @sic R5s120210 sic@

      v_UpdateType := '011'B;  // at step 21a1 the UE will send Periodic Updating

      v_UpdateResult := '100'B;

      v_UE_NetworkCap := omit; // @sic R5s120478 sic@

    } else {

      v_UpdateType := '010'B;  // at step 21a1 the UE will send combined with IMSI attach

      v_UpdateResult := '001'B;

    }

    v_RAUAccept := cs_G_RA_UpdAcc3 (cs_GMM_UpdateResult (v_UpdateResult), // @sic R5-113809, R5-120768 sic@

                                    cs_RAI (omit, cs_LAI(omit, v_PLMN, v_LAC), v_RAC ),

                                    cs_PTMSI_Signature (px_PTMSI_SigDef),

                                    omit);  // @sic R5-113809 sic@

    // Set RAU timer to 7 minutes (see 36.523-1 9.2.3.3.5 Table 9.2.3.3.5.3.3-3)

    // (TS 24.008 clause 10.5.7.4)

    v_RAUAccept.rauAccept.periodicRAupdateTimer:= cs_GprsTimer_v('001'B, '00111'B);

    //@siclog "Step 1a3" siclog@

    f_GPRS_SendDownlinkMsg(geran_Cell24, 1, cas_G_LLC_UnitData_Req (f_GERAN_TLLI_Get(geran_Cell24), tsc_LLCSapi_GMM, tsc_LLC_PM, px_GSM_CipheringOnOff, v_RAUAccept));

    //@siclog "Step 1a4 Void" siclog@

    // @sic R5-120768 sic@

    t_T3312.start;

    //@siclog "Step 1a5" siclog@

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_NonSuitableCell);

    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger); // Send trigger to EUTRA side

    // Wait for step 2 to 8 on EUTRA to complete

    f_IRAT_WaitForCoOrd_Trigger(EUTRA); // Wait trigger from  EUTRAN side

    //@siclog "Step 9a1" siclog@

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_ServingCell);

    t_T3312.timeout; // RAU can't be too early

    t_T3323.start;

    //@siclog "Step 9a2" siclog@

    v_LLCDataInd := f_GPRS_ReceiveUplinkMsg (geran_Cell24, car_G_LLC_UnitData_IND ( cr_G_RAUReq_Ids ( cr_GMM_UpdateType ( ?, '011'B),

                                                                                                      cr_RAI (omit),

                                                                                                      cr_PTMSI_SignatureAny ifpresent,

                                                                                                      ?,

                                                                                                      cdr_MobileIdTMSI_tlv (px_PTMSI_Def), // @sic R5s110607 sic@

                                                                                                      cr_MobileIdAny('1A'O) ifpresent,

                                                                                                      cr_RoutingAreaId2 ifpresent,

                                                                                                      omit))); // @sic R5s110607 sic@

    // If the timer times out before this it will be caught in the default behaviour

    t_T3323.stop;

    f_GERAN_PreliminaryPass (__FILE__, __LINE__, "Step 9a2");

    // Check ISR support

    if (v_LLCDataInd.unit.msg.rauReq.msNetworkCap.isrSupport != '1'B) {

      f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 9a2");

    }

    // @sic UE Network Cap should not be there R5-113809 sic@

    if (isvalue(v_LLCDataInd.unit.msg.rauReq.ueNetworkCapability)) {

      f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 9a2");

    }

    //@siclog "Step 9a3" siclog@

    if (v_UpdateResult == '000'B) { // This value slightly different to 1a3 and 21a3 @sic R5s120478 sic@

      v_RAUAccept.rauAccept.updateResult.ttcn_value := '100'B;

    } else {

      v_RAUAccept.rauAccept.updateResult.ttcn_value := '000'B;

    }

    f_GPRS_SendDownlinkMsg(geran_Cell24, 1, cas_G_LLC_UnitData_Req (f_GERAN_TLLI_Get(geran_Cell24), tsc_LLCSapi_GMM, tsc_LLC_PM, px_GSM_CipheringOnOff, v_RAUAccept));

    //@siclog "Step 9a4 Void" siclog@

    //@siclog "Step 9a5" siclog@

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_NonSuitableCell);

    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger); // Send trigger to EUTRA side

    // Wait for steps 10 to 20 on EUTRA to complete

    f_IRAT_WaitForCoOrd_Trigger(EUTRA); // Wait trigger from  EUTRAN side

    //@siclog "Step 21a1" siclog@

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_ServingCell);

    //@siclog "Step 21a2" siclog@

    v_LLCDataInd := f_GPRS_ReceiveUplinkMsg (geran_Cell24, car_G_LLC_UnitData_IND ( cr_G_RAUReq_Ids ( cr_GMM_UpdateType ( ?, v_UpdateType), // @sic R5s120210 sic@

                                                                                                      cr_RAI (omit),

                                                                                                      cr_PTMSI_SignatureAny ifpresent,

                                                                                                      ?,

                                                                                                      cdr_MobileIdTMSI_tlv (px_PTMSI_Def), // @sic R5s110607 sic@

                                                                                                      cr_MobileIdAny('1A'O) ifpresent,

                                                                                                      cr_RoutingAreaId2 ifpresent,

                                                                                                      v_UE_NetworkCap))); // @sic R5s110607, R5s120210 sic@

    f_GERAN_PreliminaryPass (__FILE__, __LINE__, "Step 21a2");

    // Check ISR support

    if (v_LLCDataInd.unit.msg.rauReq.msNetworkCap.isrSupport == '1'B) {

      f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 21a2");

    }

    //@siclog "Step 21a3" siclog@

    v_RAUAccept.rauAccept.updateResult.ttcn_value := v_UpdateResult;

    f_GPRS_SendDownlinkMsg(geran_Cell24, 1, cas_G_LLC_UnitData_Req (f_GERAN_TLLI_Get(geran_Cell24), tsc_LLCSapi_GMM, tsc_LLC_PM, px_GSM_CipheringOnOff, v_RAUAccept));

    //@siclog "Step 21a4 Void" siclog@

    //@siclog "Step 21a5" siclog@//

    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger); // Send trigger to EUTRA side to release cell

    // Test case now finished on GERAN side

    f_GERAN_TestBody_Set( false );

    f_GERAN_Postamble ( geran_Cell24, G1_IDLE );

  } // f_TC_9_2_3_3_5_GERAN

  


After change :

	  function f_Get_GPRS_Timer (GPRS_Timer p_gprs_timer) runs on GERAN_PTC return integer
  { 

    var integer v_Time;
    select (p_gprs_timer.unit) {
      case (tsc_GprsTimerUnit_2sec) {
        v_Time := 2 * bit2int(p_gprs_timer.timerValue);
      }
      case (tsc_GprsTimerUnit_6min) {
        v_Time := 360 * bit2int(p_gprs_timer.timerValue);
      }
      case (tsc_GprsTimerUnit_deact) {
        f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Timer not activated/started");
      }
      case else {
        v_Time := 60 * bit2int(p_gprs_timer.timerValue);
      }
    }
    return  v_Time;
  }  
…..

…..

  function f_TC_9_2_3_3_5_GERAN () runs on GERAN_PTC

  { /* Periodic Routing Area Update (multi RAT test) */

    /* Test Purpose (see f_TC_9_2_3_3_5_EUTRA) */

    /* @sic R5-113805: pc_GERAN and px_RATComb_Tested are checked at top-level sic@ */

    var octetstring v_PLMN;

    var O1_Type v_RAC;

    var O2_Type v_LAC;

    var G_LLC_DATAMESSAGE_IND v_LLCDataInd;

    var B3_Type v_UpdateType, v_UpdateResult;

    var template (value) LLCUnitDataReqType v_RAUAccept;

    var template UE_NetworkCap v_UE_NetworkCap := cr_UENetworkCap_Any ('58'O) ifpresent; // @sic R5s120210 Subject to prose CR approval sic@

    var IRAT_Coordination_MSG v_RxdMsg;

    var GERAN_SecurityInfo_Type v_SecurityInfo;

    var CoOrd_GUTIParam_Type v_GUTI;

    var bitstring v_PTMSI;
    var float v_T3314 := 44.0;
    timer t_T3323 := f_UG_SetTimerToleranceMax (7.0 * 60.0); // @sic R5-120768, R5s120210 sic@

    timer t_T3312 := f_UG_SetTimerToleranceMin (7.0 * 60.0);  // Periodic RA Update timer - 10%

    f_GERAN_Init(EUTRA_GERAN); // This is still only a 2 PTC test

    v_PLMN := f_GERAN_PLMN_Get(geran_Cell24);

    v_RAC := f_GERAN_RAC_Get(geran_Cell24);

    v_LAC := f_GERAN_LAC_Get(geran_Cell24);

    f_GERAN_CreateCell (geran_Cell24);

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);  // @sic R5s110777 sic@

    f_GERAN_Pre_Preamble (geran_Cell24, NotSpecified); // @sic R5s110176 sic@

    f_GERAN_InterRAT_InitialiseAuthParams();
    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_NonSuitableCell);

    f_GERAN_TestBody_Set( true );

    //@siclog "Step 1a1" siclog@

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_ServingCell);

    // Routing Area Update on GERAN side

    //@siclog "Step 1a2" siclog@

    v_LLCDataInd := f_GPRS_ReceiveUplinkMsg (geran_Cell24, car_G_LLC_UnitData_IND ( cr_G_RAUReq_Ids ( cr_GMM_UpdateType,  // @sic R5-113809 sic@

                                                                                                      cr_RAI (omit),

                                                                                                      cr_PTMSI_SignatureAny ifpresent,

                                                                                                      ?,

                                                                                                      omit,  // @sic R5-121764 sic@

                                                                                                      cr_MobileIdAny('1A'O) ifpresent,

                                                                                                      cr_RoutingAreaId2 ifpresent)));

    // Check ISR support

    if (v_LLCDataInd.unit.msg.rauReq.msNetworkCap.isrSupport != '1'B) {

      f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 1a2");

    }
    if ( ispresent(v_LLCDataInd.unit.msg.rauReq.readyTimer)) {
      v_T3314 := int2float(f_Get_GPRS_Timer (v_LLCDataInd.unit.msg.rauReq.readyTimer));
    }

   v_UpdateType := v_LLCDataInd.unit.msg.rauReq.updateType.updateValue;
    if (v_UpdateType == '000'B) { // @sic R5s120210 sic@

      v_UpdateType := '011'B;  // at step 21a1 the UE will send Periodic Updating

      v_UpdateResult := '100'B;

      v_UE_NetworkCap := omit; // @sic R5s120478 sic@

    } else {

      v_UpdateType := '010'B;  // at step 21a1 the UE will send combined with IMSI attach

      v_UpdateResult := '001'B;

    }
    f_GERAN_GPRS_Authentication (geran_Cell24);
    v_RAUAccept := cs_G_RA_UpdAcc3 (cs_GMM_UpdateResult (v_UpdateResult), // @sic R5-113809, R5-120768 sic@

                                    cs_RAI (omit, cs_LAI(omit, v_PLMN, v_LAC), v_RAC ),

                                    cs_PTMSI_Signature (px_PTMSI_SigDef),

                                    omit);  // @sic R5-113809 sic@

    // Set RAU timer to 7 minutes (see 36.523-1 9.2.3.3.5 Table 9.2.3.3.5.3.3-3)

    // (TS 24.008 clause 10.5.7.4)

    v_RAUAccept.rauAccept.periodicRAupdateTimer:= cs_GprsTimer_v('001'B, '00111'B);
    v_RAUAccept.rauAccept.readyTimer:= cs_GprsTimer_tv('17'O,'001'B, '00000'B);
    //@siclog "Step 1a3" siclog@

    f_GPRS_SendDownlinkMsg(geran_Cell24, 1, cas_G_LLC_UnitData_Req (f_GERAN_TLLI_Get(geran_Cell24), tsc_LLCSapi_GMM, tsc_LLC_PM, px_GSM_CipheringOnOff, v_RAUAccept));

    f_GERAN_LLC_XID (geran_Cell24, false);
    //@siclog "Step 1a4 Void" siclog@

    f_Delay (v_T3314);
    // @sic R5-120768 sic@

    t_T3312.start;

    //@siclog "Step 1a5" siclog@

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_NonSuitableCell);

    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger); // Send trigger to EUTRA side

    // Wait for step 2 to 8 on EUTRA to complete

    f_IRAT_WaitForCoOrd_Trigger(EUTRA); // Wait trigger from  EUTRAN side

    //@siclog "Step 9a1" siclog@

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_ServingCell);

    t_T3312.timeout; // RAU can't be too early

    t_T3323.start;

    //@siclog "Step 9a2" siclog@

    v_LLCDataInd := f_GPRS_ReceiveUplinkMsg (geran_Cell24, car_G_LLC_UnitData_IND ( cr_G_RAUReq_Ids ( cr_GMM_UpdateType ( ?, '011'B),

                                                                                                      cr_RAI (omit),

                                                                                                      cr_PTMSI_SignatureAny ifpresent,

                                                                                                      ?,

                                                                                                      omit, // @sic R5s110607 sic@

                                                                                                      cr_MobileIdAny('1A'O) ifpresent,

                                                                                                      cr_RoutingAreaId2 ifpresent,

                                                                                                      omit))); // @sic R5s110607 sic@

    // If the timer times out before this it will be caught in the default behaviour

    t_T3323.stop;

    f_GERAN_PreliminaryPass (__FILE__, __LINE__, "Step 9a2");

    // Check ISR support

    if (v_LLCDataInd.unit.msg.rauReq.msNetworkCap.isrSupport != '1'B) {

      f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 9a2");

    }

    // @sic UE Network Cap should not be there R5-113809 sic@

    if (isvalue(v_LLCDataInd.unit.msg.rauReq.ueNetworkCapability)) {

      f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 9a2");

    }

    //@siclog "Step 9a3" siclog@

    if (v_UpdateResult == '100'B) { // This value slightly different to 1a3 and 21a3 @sic R5s120478 sic@

      v_RAUAccept.rauAccept.updateResult.ttcn_value := '100'B;

    } else {

      v_RAUAccept.rauAccept.updateResult.ttcn_value := '000'B;

    }

    f_GPRS_SendDownlinkMsg(geran_Cell24, 1, cas_G_LLC_UnitData_Req (f_GERAN_TLLI_Get(geran_Cell24), tsc_LLCSapi_GMM, tsc_LLC_PM, px_GSM_CipheringOnOff, v_RAUAccept));

    //@siclog "Step 9a4 Void" siclog@

    //@siclog "Step 9a5" siclog@

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_NonSuitableCell);

    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger); // Send trigger to EUTRA side

    // Wait for steps 10 to 20 on EUTRA to complete

    f_IRAT_WaitForCoOrd_Trigger(EUTRA); // Wait trigger from  EUTRAN side

    //@siclog "Step 21a1" siclog@

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_ServingCell);

    //@siclog "Step 21a2" siclog@

    v_LLCDataInd := f_GPRS_ReceiveUplinkMsg (geran_Cell24, car_G_LLC_UnitData_IND ( cr_G_RAUReq_Ids ( cr_GMM_UpdateType ( ?, v_UpdateType), // @sic R5s120210 sic@

                                                                                                      cr_RAI (omit),

                                                                                                      cr_PTMSI_SignatureAny ifpresent,

                                                                                                      ?,

                                                                                                      omit, // @sic R5s110607 sic@

                                                                                                      cr_MobileIdAny('1A'O) ifpresent,

                                                                                                      cr_RoutingAreaId2 ifpresent,

                                                                                                      v_UE_NetworkCap))); // @sic R5s110607, R5s120210 sic@

    f_GERAN_PreliminaryPass (__FILE__, __LINE__, "Step 21a2");

    // Check ISR support

    if (v_LLCDataInd.unit.msg.rauReq.msNetworkCap.isrSupport != '1'B) {

      f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 21a2");

    }

    //@siclog "Step 21a3" siclog@

    v_RAUAccept.rauAccept.updateResult.ttcn_value := v_UpdateResult;

    f_GPRS_SendDownlinkMsg(geran_Cell24, 1, cas_G_LLC_UnitData_Req (f_GERAN_TLLI_Get(geran_Cell24), tsc_LLCSapi_GMM, tsc_LLC_PM, px_GSM_CipheringOnOff, v_RAUAccept));

    //@siclog "Step 21a4 Void" siclog@

    //@siclog "Step 21a5" siclog@//

    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger); // Send trigger to EUTRA side to release cell

    // Test case now finished on GERAN side

    f_GERAN_TestBody_Set( false );

    f_GERAN_Postamble ( geran_Cell24, G1_IDLE );

  } // f_TC_9_2_3_3_5_GERAN



Change 2
	Testcase name
	9.2.3.3.5

	Reason for change
	10) After RAU_ACCEPT msg , the MS performs the initial cell update by transmitting a LLC frame. 

24.008    4.7.2.1.1

If a new READY timer value is negotiated, the MS shall upon the reception of the ATTACH ACCEPT or ROUTING AREA UPDATE ACCEPT message perform a initial cell update (either by transmitting a LLC frame or, if required, a ATTACH COMPLETE or ROUTING AREA UPDATE COMPLETE message), in order to apply the new READY timer value immediately



	Summary of change
	10) Add recieiving LLC NULL frame in function  f_GERAN_LLC_XID

	Source of change
	CommonGERAN\GERAN_CommonFunctions.ttcn

	MCC160 Comment
	Not needed.  Same clause goes on to say “If the new READY timer is set to zero…..the initial cell update should not be done.”


Before change:

	  function f_GERAN_LLC_XID(GERAN_CellId_Type p_CellId,

                           boolean p_RAUComplete) runs on GERAN_PTC

  {

    var B10_Type v_BCCH_ARFCN := f_GERAN_GPRSFreq_Get (p_CellId); // @sic R5s120587 sic@

    var G_L2_DATAMESSAGE_IND v_ChanReq;

    var G_LLC_DATAMESSAGE_IND v_XID;

    timer t_Wait10S := 10.0;

    t_Wait10S.start;

    alt {

      [p_RAUComplete]  G_LLC.receive (car_G_LLC_UnitData_IND (cr_G_RA_UpdComplete))

        {

          repeat;

        }

      [] G_RLC.receive (car_G_RLC_ControlMsg_IND (p_CellId, tsc_PhyCh1, cr_PacketResourceRequestAny))

        {

          G_RLC.send (cas_G_RLC_ControlMsg_REQ_PACCH( p_CellId, tsc_PhyCh1, 1, tsc_UplinkTFI, cs_PacketUplinkAssignment(v_BCCH_ARFCN, substr(oct2bit(f_GERAN_TA_Get(p_CellId)),1, 6)))); // @sic R5s120587 sic@

          repeat;

        }

      [] G_LLC.receive (car_G_LLC_XID_IndAny)

        -> value v_XID

        {

          t_Wait10S.stop;

          // Wait for end of TBF

          // RACE CONDITION: Do we need to check the flag here ???

          G_RLC.receive (car_G_RLC_ControlMsg_IND(p_CellId, tsc_PhyCh1, cr_PacketControlAcknowledgement)); // RACE CONDITION

          fl_SendXIDRes (p_CellId, v_XID.xid.sAPI);

        }

      [] G_RLC.receive (car_G_RLC_ControlMsg_IND(p_CellId, tsc_PhyCh1, cr_PacketControlAcknowledgement))

        {

          // RACE CONDITION: Do we need to set a flag here ???

          repeat;

        }

      [] G_L2.receive (car_G_L2_ACCESS_IND ( p_CellId , tsc_PhyCh0, tsc_RACH, ?, ?, cr_ChanReqOnePhase ))

        -> value v_ChanReq

        {

          fl_ULTBFOnePhase (p_CellId, v_ChanReq.access); // @sic R5s120050 MCC160 implementation sic@

          repeat;

        }

      [] t_Wait10S.timeout {} // No XID coming

    }

  } //end of f_GERAN_LLC_XID


After change :

	  function f_GERAN_LLC_XID(GERAN_CellId_Type p_CellId,

                           boolean p_RAUComplete) runs on GERAN_PTC

  {

    var B10_Type v_BCCH_ARFCN := f_GERAN_GPRSFreq_Get (p_CellId); // @sic R5s120587 sic@

    var G_L2_DATAMESSAGE_IND v_ChanReq;

    var G_LLC_DATAMESSAGE_IND v_XID;

    timer t_Wait10S := 10.0;

    t_Wait10S.start;

    alt {

      [] G_LLC.receive (car_G_LLC_NULL_Ind)

        {

          repeat;

        }

      [p_RAUComplete]  G_LLC.receive (car_G_LLC_UnitData_IND (cr_G_RA_UpdComplete))

        {

          repeat;

        }

      [] G_RLC.receive (car_G_RLC_ControlMsg_IND (p_CellId, tsc_PhyCh1, cr_PacketResourceRequestAny))

        {

          G_RLC.send (cas_G_RLC_ControlMsg_REQ_PACCH( p_CellId, tsc_PhyCh1, 1, tsc_UplinkTFI, cs_PacketUplinkAssignment(v_BCCH_ARFCN, substr(oct2bit(f_GERAN_TA_Get(p_CellId)),1, 6)))); // @sic R5s120587 sic@

          repeat;

        }

      [] G_LLC.receive (car_G_LLC_XID_IndAny)

        -> value v_XID

        {

          t_Wait10S.stop;

          // Wait for end of TBF

          // RACE CONDITION: Do we need to check the flag here ???

          G_RLC.receive (car_G_RLC_ControlMsg_IND(p_CellId, tsc_PhyCh1, cr_PacketControlAcknowledgement)); // RACE CONDITION

          fl_SendXIDRes (p_CellId, v_XID.xid.sAPI);

        }

      [] G_RLC.receive (car_G_RLC_ControlMsg_IND(p_CellId, tsc_PhyCh1, cr_PacketControlAcknowledgement))

        {

          // RACE CONDITION: Do we need to set a flag here ???

          repeat;

        }

      [] G_L2.receive (car_G_L2_ACCESS_IND ( p_CellId , tsc_PhyCh0, tsc_RACH, ?, ?, cr_ChanReqOnePhase ))

        -> value v_ChanReq

        {

          fl_ULTBFOnePhase (p_CellId, v_ChanReq.access); // @sic R5s120050 MCC160 implementation sic@

          repeat;

        }

      [] t_Wait10S.timeout {} // No XID coming

    }

  } //end of f_GERAN_LLC_XID

